```
> restart;
  with(RootFinding);
  with(plots);
```
$$[Analytic, AnalyticZerosFound, BivariatePolynomial, Homotopy, Isolate, NextZero, Parametric] \quad (1)$$

$$[animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, densityplot, display, dualaxisplot, fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, intersectplot, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra\_supported, polyhedraplot, rootlocus, semilogplot, setcolors, setoptions, setoptions3d, spacecurve, sparsematrixplot, surfdata, textplot, textplot3d, tubeplot]$$

Adams-Bashford algorithms

The purpose of this worksheet is to derive the (N+1)-stage Adams-Bashford algorithms (N < 4) for the solution of

```
> ode := diff(y(x),x) = f(x,y(x));
```
$$ode := \frac{d}{dx} y(x) = f(x, y(x)) \quad (2)$$

This algorithm is defined by

```
> N := 3;
  eq1 := y(x+h) = y(x) + h*add(beta[i]*f(x-i*h,y(x-i*h)),i=0..N);
```
$$N := 3 \quad (3)$$

$$eq1 := y(x+h) = y(x) + h\left(\beta_0 f(x, y(x)) + \beta_1 f(x-h, y(x-h)) + \beta_2 f(x-2h, y(x-2h)) + \beta_3 f(x-3h, y(x-3h))\right)$$

The idea is to choose the b[i] coefficients such that the RHS gives an approximation to the LHS accurate to order h^(N+1).

```
> eq2 := series((lhs-rhs)(eq1),h,N+2);
```
$$eq2 := \left(D(y)(x) - \beta_0 f(x, y(x)) - \beta_1 f(x, y(x)) - \beta_2 f(x, y(x)) - \beta_3 f(x, y(x))\right) h \quad (4)$$

$$+ \left(-\beta_1 \left(-D_1(f)(x, y(x)) - D_2(f)(x, y(x))\, D(y)(x)\right) - \beta_2 \left(-2\, D_1(f)(x, y(x)) - 2\, D_2(f)(x, y(x))\, D(y)(x)\right) - \beta_3 \left(-3\, D_1(f)(x, y(x)) - 3\, D_2(f)(x, y(x))\, D(y)(x)\right) + \frac{1}{2} D^{(2)}(y)(x)\right) h^2 + \left(-\beta_1 \left(\frac{1}{2} D_{1,1}(f)(x, y(x))\right.\right.$$

$$+ D_{1,2}(f)(x, y(x))\, D(y)(x) + \frac{1}{2} D_{2,2}(f)(x, y(x))\, D(y)(x)^2 + \frac{1}{2} D_2(f)(x,$$

$$y(x))\,\mathrm{D}^{(2)}(y)(x)\Big) - \beta_2\,\Big(2\,\mathrm{D}_{1,1}(f)\,(x, y(x)) + 4\,\mathrm{D}_{1,2}(f)\,(x, y(x))\,\mathrm{D}(y)(x)$$

$$+ 2\,\mathrm{D}_{2,2}(f)\,(x, y(x))\,\mathrm{D}(y)(x)^2 + 2\,\mathrm{D}_2(f)\,(x, y(x))\,\mathrm{D}^{(2)}(y)(x)\Big)$$

$$- \beta_3\,\Big(\frac{9}{2}\,\mathrm{D}_{1,1}(f)\,(x, y(x)) + 9\,\mathrm{D}_{1,2}(f)\,(x, y(x))\,\mathrm{D}(y)(x) + \frac{9}{2}\,\mathrm{D}_{2,2}(f)\,(x,$$

$$y(x))\,\mathrm{D}(y)(x)^2 + \frac{9}{2}\,\mathrm{D}_2(f)\,(x, y(x))\,\mathrm{D}^{(2)}(y)(x)\Big) + \frac{1}{6}\,\mathrm{D}^{(3)}(y)(x)\Big)\,h^3$$

$$+ \Big(\frac{1}{24}\,\mathrm{D}^{(4)}(y)(x) - \beta_1\,\Big(-\frac{1}{6}\,\mathrm{D}_{1,1,1}(f)\,(x, y(x)) - \frac{1}{2}\,\mathrm{D}_{1,1,2}(f)\,(x,$$

$$y(x))\,\mathrm{D}(y)(x) - \frac{1}{2}\,\mathrm{D}_{1,2,2}(f)\,(x, y(x))\,\mathrm{D}(y)(x)^2 - \frac{1}{2}\,\mathrm{D}_{1,2}(f)\,(x,$$

$$y(x))\,\mathrm{D}^{(2)}(y)(x) - \frac{1}{6}\,\mathrm{D}(y)(x)^3\,\mathrm{D}_{2,2,2}(f)\,(x, y(x)) - \frac{1}{2}\,\mathrm{D}_{2,2}(f)\,(x,$$

$$y(x))\,\mathrm{D}(y)(x)\,\mathrm{D}^{(2)}(y)(x) - \frac{1}{6}\,\mathrm{D}_2(f)\,(x, y(x))\,\mathrm{D}^{(3)}(y)(x)\Big) - \beta_2\,\Big($$

$$-\frac{4}{3}\,\mathrm{D}_{1,1,1}(f)\,(x, y(x)) - 4\,\mathrm{D}_{1,1,2}(f)\,(x, y(x))\,\mathrm{D}(y)(x) - 4\,\mathrm{D}_{1,2,2}(f)\,(x,$$

$$y(x))\,\mathrm{D}(y)(x)^2 - 4\,\mathrm{D}_{1,2}(f)\,(x, y(x))\,\mathrm{D}^{(2)}(y)(x) - \frac{4}{3}\,\mathrm{D}(y)(x)^3\,\mathrm{D}_{2,2,2}(f)\,(x,$$

$$y(x)) - 4\,\mathrm{D}_{2,2}(f)\,(x, y(x))\,\mathrm{D}(y)(x)\,\mathrm{D}^{(2)}(y)(x) - \frac{4}{3}\,\mathrm{D}_2(f)\,(x,$$

$$y(x))\,\mathrm{D}^{(3)}(y)(x)\Big) - \beta_3\,\Big(-\frac{9}{2}\,\mathrm{D}_{1,1,1}(f)\,(x, y(x)) - \frac{27}{2}\,\mathrm{D}_{1,1,2}(f)\,(x,$$

$$y(x))\,\mathrm{D}(y)(x) - \frac{27}{2}\,\mathrm{D}_{1,2,2}(f)\,(x, y(x))\,\mathrm{D}(y)(x)^2 - \frac{27}{2}\,\mathrm{D}_{1,2}(f)\,(x,$$

$$y(x))\,\mathrm{D}^{(2)}(y)(x) - \frac{9}{2}\,\mathrm{D}(y)(x)^3\,\mathrm{D}_{2,2,2}(f)\,(x, y(x)) - \frac{27}{2}\,\mathrm{D}_{2,2}(f)\,(x,$$

$$y(x))\,\mathrm{D}(y)(x)\,\mathrm{D}^{(2)}(y)(x) - \frac{9}{2}\,\mathrm{D}_2(f)\,(x, y(x))\,\mathrm{D}^{(3)}(y)(x)\Big)\Big)\,h^4 + \mathrm{O}(h^5)$$

If you look at eq2, you'll see terms like D(y)(x), D(2)(y)(x), etc. These can all be expressed in terms of f using the original ODE:

```
> derivative[1] := convert(ode,D);
  for i from 2 to N+1 do:
     derivative[i] := subs(seq(derivative[j],j=1..i-1),convert(diff
  (derivative[i-1],x),D)):
  od;
```

$$derivative_1 := \mathrm{D}(y)(x) = f(x, y(x)) \tag{5}$$

$$derivative_2 := D^{(2)}(y)(x) = D_1(f)(x, y(x)) + D_2(f)(x, y(x))\, f(x, y(x))$$

$$derivative_3 := D^{(3)}(y)(x) = D_{1,1}(f)(x, y(x)) + D_{1,2}(f)(x, y(x))\, f(x, y(x)) + \left(D_{1,2}(f)(x, y(x)) + D_{2,2}(f)(x, y(x))\, f(x, y(x))\right) f(x, y(x)) + D_2(f)(x, y(x)) \left(D_1(f)(x, y(x)) + D_2(f)(x, y(x))\, f(x, y(x))\right)$$

$$derivative_4 := D^{(4)}(y)(x) = D_{1,1,1}(f)(x, y(x)) + D_{1,1,2}(f)(x, y(x))\, f(x, y(x)) + \left(D_{1,1,2}(f)(x, y(x)) + D_{1,2,2}(f)(x, y(x))\, f(x, y(x))\right) f(x, y(x)) + D_{1,2}(f)(x, y(x)) \left(D_1(f)(x, y(x)) + D_2(f)(x, y(x))\, f(x, y(x))\right) + \left(D_{1,1,2}(f)(x, y(x)) + D_{1,2,2}(f)(x, y(x))\, f(x, y(x)) + \left(D_{1,2,2}(f)(x, y(x)) + D_{2,2,2}(f)(x, y(x))\, f(x, y(x))\right) f(x, y(x)) + D_{2,2}(f)(x, y(x)) \left(D_1(f)(x, y(x)) + D_2(f)(x, y(x))\, f(x, y(x))\right)\right) f(x, y(x)) + 2 \left(D_{1,2}(f)(x, y(x)) + D_{2,2}(f)(x, y(x))\, f(x, y(x))\right) \left(D_1(f)(x, y(x)) + D_2(f)(x, y(x))\, f(x, y(x))\right) + D_2(f)(x, y(x)) \left(D_{1,1}(f)(x, y(x)) + D_{1,2}(f)(x, y(x))\, f(x, y(x)) + \left(D_{1,2}(f)(x, y(x)) + D_{2,2}(f)(x, y(x))\, f(x, y(x))\right) f(x, y(x)) + D_2(f)(x, y(x)) \left(D_1(f)(x, y(x)) + D_2(f)(x, y(x))\, f(x, y(x))\right)\right)$$

Subbing these derivatives into eq2 yields:

```
> eq3 := subs(convert(derivative,list)$N,eq2);
```

$$eq3 := \left(f(x, y(x)) - \beta_0 f(x, y(x)) - \beta_1 f(x, y(x)) - \beta_2 f(x, y(x)) - \beta_3 f(x, y(x))\right) h + \left(-\beta_1 \left(-D_1(f)(x, y(x)) - D_2(f)(x, y(x))\, f(x, y(x))\right) - \beta_2 \left(-2\, D_1(f)(x, y(x)) - 2\, D_2(f)(x, y(x))\, f(x, y(x))\right) - \beta_3 \left(-3\, D_1(f)(x, y(x)) - 3\, D_2(f)(x, y(x))\, f(x, y(x))\right) + \frac{1}{2} D_1(f)(x, y(x)) + \frac{1}{2} D_2(f)(x, y(x))\, f(x, y(x))\right) h^2 + \left(-\beta_1 \left(\frac{1}{2} D_{1,1}(f)(x, y(x)) + D_{1,2}(f)(x, y(x))\, f(x, y(x)) + \frac{1}{2} D_{2,2}(f)(x, y(x))\, f(x, y(x))^2 + \frac{1}{2} D_2(f)(x, y(x)) \left(D_1(f)(x, y(x)) + D_2(f)(x, y(x))\, f(x, y(x))\right)\right) - \beta_2 \left(2\, D_{1,1}(f)(x, y(x)) + 4\, D_{1,2}(f)(x, y(x))\, f(x, y(x)) + 2\, D_{2,2}(f)(x, y(x))\, f(x, y(x))^2 + 2\, D_2(f)(x, y(x)) \left(D_1(f)(x, y(x))\right.\right.\right. \tag{6}$$

$$+ D_2(f)(x, y(x)) f(x, y(x)))\Big) - \beta_3 \Big(\frac{9}{2} D_{1,1}(f)(x, y(x)) + 9 D_{1,2}(f)(x,$$

$$y(x)) f(x, y(x)) + \frac{9}{2} D_{2,2}(f)(x, y(x)) f(x, y(x))^2 + \frac{9}{2} D_2(f)(x,$$

$$y(x)) (D_1(f)(x, y(x)) + D_2(f)(x, y(x)) f(x, y(x)))\Big) + \frac{1}{6} D_{1,1}(f)(x, y(x))$$

$$+ \frac{1}{6} D_{1,2}(f)(x, y(x)) f(x, y(x)) + \frac{1}{6} (D_{1,2}(f)(x, y(x)) + D_{2,2}(f)(x,$$

$$y(x)) f(x, y(x))) f(x, y(x)) + \frac{1}{6} D_2(f)(x, y(x)) (D_1(f)(x, y(x))$$

$$+ D_2(f)(x, y(x)) f(x, y(x)))\Big) h^3 + \Big(\frac{1}{24} D_{1,1,1}(f)(x, y(x))$$

$$+ \frac{1}{24} D_{1,1,2}(f)(x, y(x)) f(x, y(x)) + \frac{1}{24} (D_{1,1,2}(f)(x, y(x))$$

$$+ D_{1,2,2}(f)(x, y(x)) f(x, y(x))) f(x, y(x)) + \frac{1}{24} D_{1,2}(f)(x,$$

$$y(x)) (D_1(f)(x, y(x)) + D_2(f)(x, y(x)) f(x, y(x))) + \frac{1}{24} (D_{1,1,2}(f)(x,$$

$$y(x)) + D_{1,2,2}(f)(x, y(x)) f(x, y(x)) + (D_{1,2,2}(f)(x, y(x)) + D_{2,2,2}(f)(x,$$

$$y(x)) f(x, y(x))) f(x, y(x)) + D_{2,2}(f)(x, y(x)) (D_1(f)(x, y(x)) + D_2(f)(x,$$

$$y(x)) f(x, y(x)))) f(x, y(x)) + \frac{1}{12} (D_{1,2}(f)(x, y(x)) + D_{2,2}(f)(x,$$

$$y(x)) f(x, y(x))) (D_1(f)(x, y(x)) + D_2(f)(x, y(x)) f(x, y(x)))$$

$$+ \frac{1}{24} D_2(f)(x, y(x)) (D_{1,1}(f)(x, y(x)) + D_{1,2}(f)(x, y(x)) f(x, y(x))$$

$$+ (D_{1,2}(f)(x, y(x)) + D_{2,2}(f)(x, y(x)) f(x, y(x))) f(x, y(x)) + D_2(f)(x,$$

$$y(x)) (D_1(f)(x, y(x)) + D_2(f)(x, y(x)) f(x, y(x)))) - \beta_1 \Big($$

$$-\frac{1}{6} D_{1,1,1}(f)(x, y(x)) - \frac{1}{2} D_{1,1,2}(f)(x, y(x)) f(x, y(x)) - \frac{1}{2} D_{1,2,2}(f)(x,$$

$$y(x)) f(x, y(x))^2 - \frac{1}{2} D_{1,2}(f)(x, y(x)) (D_1(f)(x, y(x)) + D_2(f)(x,$$

$$y(x))\, f(x, y(x))) - \frac{1}{6} f(x, y(x))^{3}\, \mathrm{D}_{2, 2, 2}(f)\,(x, y(x)) - \frac{1}{2} \mathrm{D}_{2, 2}(f)\,(x,$$

$$y(x))\, f(x, y(x))\, (\mathrm{D}_{1}(f)\,(x, y(x)) + \mathrm{D}_{2}(f)\,(x, y(x))\, f(x, y(x)))$$

$$- \frac{1}{6} \mathrm{D}_{2}(f)\,(x, y(x))\, (\mathrm{D}_{1, 1}(f)\,(x, y(x)) + \mathrm{D}_{1, 2}(f)\,(x, y(x))\, f(x, y(x))$$

$$+ (\mathrm{D}_{1, 2}(f)\,(x, y(x)) + \mathrm{D}_{2, 2}(f)\,(x, y(x))\, f(x, y(x)))\, f(x, y(x)) + \mathrm{D}_{2}(f)\,(x,$$

$$y(x))\, (\mathrm{D}_{1}(f)\,(x, y(x)) + \mathrm{D}_{2}(f)\,(x, y(x))\, f(x, y(x))))) - \beta_{2}\, ($$

$$- \frac{4}{3} \mathrm{D}_{1, 1, 1}(f)\,(x, y(x)) - 4\, \mathrm{D}_{1, 1, 2}(f)\,(x, y(x))\, f(x, y(x)) - 4\, \mathrm{D}_{1, 2, 2}(f)\,(x,$$

$$y(x))\, f(x, y(x))^{2} - 4\, \mathrm{D}_{1, 2}(f)\,(x, y(x))\, (\mathrm{D}_{1}(f)\,(x, y(x)) + \mathrm{D}_{2}(f)\,(x,$$

$$y(x))\, f(x, y(x))) - \frac{4}{3} f(x, y(x))^{3}\, \mathrm{D}_{2, 2, 2}(f)\,(x, y(x)) - 4\, \mathrm{D}_{2, 2}(f)\,(x,$$

$$y(x))\, f(x, y(x))\, (\mathrm{D}_{1}(f)\,(x, y(x)) + \mathrm{D}_{2}(f)\,(x, y(x))\, f(x, y(x)))$$

$$- \frac{4}{3} \mathrm{D}_{2}(f)\,(x, y(x))\, (\mathrm{D}_{1, 1}(f)\,(x, y(x)) + \mathrm{D}_{1, 2}(f)\,(x, y(x))\, f(x, y(x))$$

$$+ (\mathrm{D}_{1, 2}(f)\,(x, y(x)) + \mathrm{D}_{2, 2}(f)\,(x, y(x))\, f(x, y(x)))\, f(x, y(x)) + \mathrm{D}_{2}(f)\,(x,$$

$$y(x))\, (\mathrm{D}_{1}(f)\,(x, y(x)) + \mathrm{D}_{2}(f)\,(x, y(x))\, f(x, y(x))))) - \beta_{3}\, ($$

$$- \frac{9}{2} \mathrm{D}_{1, 1, 1}(f)\,(x, y(x)) - \frac{27}{2} \mathrm{D}_{1, 1, 2}(f)\,(x, y(x))\, f(x, y(x))$$

$$- \frac{27}{2} \mathrm{D}_{1, 2, 2}(f)\,(x, y(x))\, f(x, y(x))^{2} - \frac{27}{2} \mathrm{D}_{1, 2}(f)\,(x, y(x))\, (\mathrm{D}_{1}(f)\,(x,$$

$$y(x)) + \mathrm{D}_{2}(f)\,(x, y(x))\, f(x, y(x))) - \frac{9}{2} f(x, y(x))^{3}\, \mathrm{D}_{2, 2, 2}(f)\,(x, y(x))$$

$$- \frac{27}{2} \mathrm{D}_{2, 2}(f)\,(x, y(x))\, f(x, y(x))\, (\mathrm{D}_{1}(f)\,(x, y(x)) + \mathrm{D}_{2}(f)\,(x, y(x))\, f(x,$$

$$y(x))) - \frac{9}{2} \mathrm{D}_{2}(f)\,(x, y(x))\, (\mathrm{D}_{1, 1}(f)\,(x, y(x)) + \mathrm{D}_{1, 2}(f)\,(x, y(x))\, f(x, y(x))$$

$$+ (\mathrm{D}_{1, 2}(f)\,(x, y(x)) + \mathrm{D}_{2, 2}(f)\,(x, y(x))\, f(x, y(x)))\, f(x, y(x)) + \mathrm{D}_{2}(f)\,(x,$$

$$y(x))\, (\mathrm{D}_{1}(f)\,(x, y(x)) + \mathrm{D}_{2}(f)\,(x, y(x))\, f(x, y(x)))))))\, h^{4} + \mathrm{O}(h^{5})$$

This is a polynomial in the following variables:

```
> vars := [h,f(x,y(x)),D[1](f)(x,y(x)),D[2](f)(x,y(x)),D[1,1](f)(x,y(x)),D[1,2](f)(x,y(x)),D[2,2](f)(x,y(x)),D[1,1,1](f)(x,y(x)),D[1,1,2](f)(x,y(x)),D[1,2,2](f)(x,y(x)),D[2,2,2](f)(x,y(x))];
```

$$vars := \left[h, f(x,y(x)), \mathrm{D}_1(f)(x,y(x)), \mathrm{D}_2(f)(x,y(x)), \mathrm{D}_{1,1}(f)(x,y(x)), \mathrm{D}_{1,2}(f)(x,y(x)), \mathrm{D}_{2,2}(f)(x,y(x)), \mathrm{D}_{1,1,1}(f)(x,y(x)), \mathrm{D}_{1,1,2}(f)(x,y(x)), \mathrm{D}_{1,2,2}(f)(x,y(x)), \mathrm{D}_{2,2,2}(f)(x,y(x))\right] \tag{7}$$

We can get a better looking expression for eq3 by using the collect command to group terms with the same powers of the quantities in vars. To get this to work properly, we need to first convert eq3 from a series into a polynomial (essentially dropping the O(h^(N+2)) term):

```
> eq4 := collect(convert(eq3,polynom),vars,'distributed');
```

$$\begin{aligned} eq4 := {} & \left(\frac{1}{6}\beta_1 + \frac{4}{3}\beta_2 + \frac{1}{24} + \frac{9}{2}\beta_3\right)\mathrm{D}_{2,2,2}(f)(x,y(x))\,h^4 f(x,y(x))^3 + \left(\frac{16}{3}\beta_2 \right. \\ & \left. + \frac{2}{3}\beta_1 + \frac{1}{6} + 18\,\beta_3\right)\mathrm{D}_{2,2}(f)(x,y(x))\,h^4 f(x,y(x))^2\,\mathrm{D}_2(f)(x,y(x)) + \left(4\,\beta_2 \right. \\ & \left. + \frac{27}{2}\beta_3 + \frac{1}{8} + \frac{1}{2}\beta_1\right)\mathrm{D}_{1,2,2}(f)(x,y(x))\,h^4 f(x,y(x))^2 + \left(4\,\beta_2 + \frac{27}{2}\beta_3 \right. \\ & \left. + \frac{1}{8} + \frac{1}{2}\beta_1\right)\mathrm{D}_{2,2}(f)(x,y(x))\,h^4 f(x,y(x))\,\mathrm{D}_1(f)(x,y(x)) + \left(\frac{1}{6}\beta_1 \right. \\ & \left. + \frac{4}{3}\beta_2 + \frac{1}{24} + \frac{9}{2}\beta_3\right)h^4 f(x,y(x))\,\mathrm{D}_2(f)(x,y(x))^3 + \left(\frac{5}{24} + \frac{5}{6}\beta_1 + \frac{20}{3}\beta_2 \right. \\ & \left. + \frac{45}{2}\beta_3\right)\mathrm{D}_{1,2}(f)(x,y(x))\,h^4 f(x,y(x))\,\mathrm{D}_2(f)(x,y(x)) + \left(4\,\beta_2 + \frac{27}{2}\beta_3 \right. \\ & \left. + \frac{1}{8} + \frac{1}{2}\beta_1\right)\mathrm{D}_{1,1,2}(f)(x,y(x))\,h^4 f(x,y(x)) + \left(\frac{1}{6}\beta_1 + \frac{4}{3}\beta_2 + \frac{1}{24} \right. \\ & \left. + \frac{9}{2}\beta_3\right)h^4\,\mathrm{D}_1(f)(x,y(x))\,\mathrm{D}_2(f)(x,y(x))^2 + \left(4\,\beta_2 + \frac{27}{2}\beta_3 + \frac{1}{8} \right. \\ & \left. + \frac{1}{2}\beta_1\right)\mathrm{D}_{1,2}(f)(x,y(x))\,h^4\,\mathrm{D}_1(f)(x,y(x)) + \left(\frac{1}{6}\beta_1 + \frac{4}{3}\beta_2 + \frac{1}{24} \right. \\ & \left. + \frac{9}{2}\beta_3\right)\mathrm{D}_{1,1}(f)(x,y(x))\,h^4\,\mathrm{D}_2(f)(x,y(x)) + \left(\frac{1}{6}\beta_1 + \frac{4}{3}\beta_2 + \frac{1}{24} \right. \\ & \left. + \frac{9}{2}\beta_3\right)\mathrm{D}_{1,1,1}(f)(x,y(x))\,h^4 + \left(-\frac{9}{2}\beta_3 + \frac{1}{6} - 2\,\beta_2 - \frac{1}{2}\beta_1\right)\mathrm{D}_{2,2}(f)(x, \\ & y(x))\,h^3 f(x,y(x))^2 + \left(-\frac{9}{2}\beta_3 + \frac{1}{6} - 2\,\beta_2 - \frac{1}{2}\beta_1\right)h^3 f(x,y(x))\,\mathrm{D}_2(f)(x, \\ & y(x))^2 + \left(-4\,\beta_2 - \beta_1 + \frac{1}{3} - 9\,\beta_3\right)\mathrm{D}_{1,2}(f)(x,y(x))\,h^3 f(x,y(x)) + \left(-\frac{9}{2}\beta_3 \right. \end{aligned}$$

$$+\frac{1}{6}-2\beta_2-\frac{1}{2}\beta_1\Big)\mathrm{D}_2(f)(x,y(x))\,h^3\,\mathrm{D}_1(f)(x,y(x))+\Big(-\frac{9}{2}\beta_3+\frac{1}{6}-2\beta_2-\frac{1}{2}\beta_1\Big)\mathrm{D}_{1,1}(f)(x,y(x))\,h^3+\Big(\beta_1+2\beta_2+3\beta_3+\frac{1}{2}\Big)\mathrm{D}_2(f)(x,y(x))\,h^2 f(x,y(x))+\Big(\beta_1+2\beta_2+3\beta_3+\frac{1}{2}\Big)\mathrm{D}_1(f)(x,y(x))\,h^2+\big(1-\beta_0-\beta_1-\beta_2-\beta_3\big)f(x,y(x))\,h$$

We then extract the coefficients of the polynomial and put them into a list:

```
> eq6 := [coeffs(eq4,vars)];
```

$$eq6:=\Big[\frac{1}{6}\beta_1+\frac{4}{3}\beta_2+\frac{1}{24}+\frac{9}{2}\beta_3,\ \frac{1}{6}\beta_1+\frac{4}{3}\beta_2+\frac{1}{24}+\frac{9}{2}\beta_3,\ -\frac{9}{2}\beta_3+\frac{1}{6}-2\beta_2-\frac{1}{2}\beta_1,\ \frac{1}{6}\beta_1+\frac{4}{3}\beta_2+\frac{1}{24}+\frac{9}{2}\beta_3,\ -\frac{9}{2}\beta_3+\frac{1}{6}-2\beta_2-\frac{1}{2}\beta_1,\ \beta_1+2\beta_2+3\beta_3+\frac{1}{2},\ 1-\beta_0-\beta_1-\beta_2-\beta_3,\ \frac{16}{3}\beta_2+\frac{2}{3}\beta_1+\frac{1}{6}+18\beta_3,\ \frac{1}{6}\beta_1+\frac{4}{3}\beta_2+\frac{1}{24}+\frac{9}{2}\beta_3,\ 4\beta_2+\frac{27}{2}\beta_3+\frac{1}{8}+\frac{1}{2}\beta_1,\ 4\beta_2+\frac{27}{2}\beta_3+\frac{1}{8}+\frac{1}{2}\beta_1,\ \frac{5}{24}+\frac{5}{6}\beta_1+\frac{20}{3}\beta_2+\frac{45}{2}\beta_3,\ 4\beta_2+\frac{27}{2}\beta_3+\frac{1}{8}+\frac{1}{2}\beta_1,\ 4\beta_2+\frac{27}{2}\beta_3+\frac{1}{8}+\frac{1}{2}\beta_1,\ \frac{1}{6}\beta_1+\frac{4}{3}\beta_2+\frac{1}{24}+\frac{9}{2}\beta_3,\ -\frac{9}{2}\beta_3+\frac{1}{6}-2\beta_2-\frac{1}{2}\beta_1,\ -4\beta_2-\beta_1+\frac{1}{3}-9\beta_3,\ -\frac{9}{2}\beta_3+\frac{1}{6}-2\beta_2-\frac{1}{2}\beta_1,\ \beta_1+2\beta_2+3\beta_3+\frac{1}{2}\Big] \tag{9}$$

This is a simple linear system for (N+1) equations for the (N+1) b[i] coefficients which can be solved as follows:

```
> eq7 := solve(eq6);
```

$$eq7:=\Big\{\beta_0=\frac{55}{24},\ \beta_1=-\frac{59}{24},\ \beta_2=\frac{37}{24},\ \beta_3=-\frac{3}{8}\Big\} \tag{10}$$

Putting our solution into eq1, we get a stencil for y(x+h) accurate to order h^(N+1):

```
> eq8 := collect(subs(eq7,eq1),h);
```

$$eq8:=y(x+h)=y(x)+h\Big(\frac{55}{24}f(x,y(x))-\frac{59}{24}f(x-h,y(x-h))+\frac{37}{24}f(x-2h,y(x-2h))-\frac{3}{8}f(x-3h,y(x-3h))\Big) \tag{11}$$

To verify the accuracy, we expand the (lhs-rhs) of eq8 in a series and use the derivative expressions above

```
> simplify(subs(convert(derivative,list),series((lhs-rhs)(eq8),h,
  N+3)));
```

$$\tag{12}$$

$$\left(\frac{49}{24}f(x,y(x))^2 D_{2,2,2}(f)(x,y(x))D_1(f)(x,y(x))+\frac{49}{12}D_{1,2,2}(f)(x,y(x))f(x,y(x))D_1(f)(x,y(x))+\frac{343}{144}f(x,y(x))^3 D_{2,2,2}(f)(x,y(x))D_2(f)(x,y(x))+\frac{539}{144}D_{2,2}(f)(x,y(x))f(x,y(x))^2 D_2(f)(x,y(x))^2+\frac{49}{36}D_{2,2}(f)(x,y(x))f(x,y(x))D_{1,1}(f)(x,y(x))+\frac{49}{16}D_{1,2}(f)(x,y(x))D_2(f)(x,y(x))^2 f(x,y(x))+\frac{343}{144}D_{1,2}(f)(x,y(x))D_2(f)(x,y(x))D_1(f)(x,y(x))+\frac{49}{16}D_{1,1,2}(f)(x,y(x))D_2(f)(x,y(x))f(x,y(x))+\frac{245}{48}D_{1,2,2}(f)(x,y(x))f(x,y(x))^2 D_2(f)(x,y(x))+\frac{49}{12}D_{2,2}(f)(x,y(x))f(x,y(x))^2 D_{1,2}(f)(x,y(x))+\frac{637}{144}D_{2,2}(f)(x,y(x))f(x,y(x))D_2(f)(x,y(x))D_1(f)(x,y(x))+\frac{49}{144}D_{1,1,1,1}(f)(x,y(x))+\frac{1}{120}D^{(5)}(y)(x)+\frac{49}{24}D_{1,1,2}(f)(x,y(x))D_1(f)(x,y(x))+\frac{49}{48}D_{2,2}(f)(x,y(x))D_1(f)(x,y(x))^2+\frac{49}{36}D_{1,1,1,2}(f)(x,y(x))f(x,y(x))+\frac{49}{24}D_{1,1,2,2}(f)(x,y(x))f(x,y(x))^2+\frac{49}{36}f(x,y(x))^3 D_{1,2,2,2}(f)(x,y(x))+\frac{49}{144}f(x,y(x))^4 D_{2,2,2,2}(f)(x,y(x))+\frac{49}{36}D_{2,2}(f)(x,y(x))^2 f(x,y(x))^3+\frac{49}{18}D_{1,2}(f)(x,y(x))^2 f(x,y(x))+\frac{49}{36}D_{1,2}(f)(x,y(x))D_{1,1}(f)(x,y(x))+\frac{49}{144}D_2(f)(x,y(x))^4 f(x,y(x))+\frac{49}{144}D_2(f)(x,y(x))^2 D_{1,1}(f)(x,y(x))+\frac{49}{144}D_2(f)(x,y(x))^3 D_1(f)(x,y(x))+\frac{49}{144}D_2(f)(x,y(x))D_{1,1,1}(f)(x,y(x))\right)h^5+O\left(h^6\right) \tag{12}$$

As expected, the one-step error is O(h^(N+2)). Now, let us specialize to the test problem y' = lambda*y:

```
> f := (x,y) -> lambda*y;
  ode;
  eq8;
```

$$f := (x, y) \to \lambda\, y \tag{13}$$

$$\frac{d}{dx}\, y(x) = \lambda\, y(x)$$

$$y(x+h) = y(x) + h\left(\frac{55}{24}\lambda\, y(x) - \frac{59}{24}\lambda\, y(x-h) + \frac{37}{24}\lambda\, y(x-2h) - \frac{3}{8}\lambda\, y(x - 3h)\right)$$

As discussed in class, the error in our numeric approximation for the n^th value of y is written as E(n). The E(n) satisfy eq8 with the y(x+i*h) = E(n+i):

```
> Subs := [seq(y(x+i*h)=E(n+i),i=-N..1),h=z/lambda];
  eq9 := (lhs-rhs)(expand(subs(Subs,eq8)))=0;
```

$$Subs := \left[y(x-3h) = E(n-3), y(x-2h) = E(n-2), y(x-h) = E(n-1), y(x) = E(n), y(x+h) = E(n+1), h = \frac{z}{\lambda}\right] \tag{14}$$

$$eq9 := E(n+1) - E(n) - \frac{55}{24} z\, E(n) + \frac{59}{24} z\, E(n-1) - \frac{37}{24} z\, E(n-2) + \frac{3}{8} z\, E(n-3) = 0$$

Above, we have also written z=lambda*h. We make the ansatz E(n) = xi^n, which re-writes eq9 as

```
> E := n -> xi^n;
  eq10 := collect(simplify(eq9/xi^(n-N)),xi);
```

$$E := n \to \xi^n \tag{15}$$

$$eq10 := \xi^4 + \left(-\frac{55 z}{24} - 1\right)\xi^3 + \frac{59\, \xi^2 z}{24} - \frac{37\, \xi\, z}{24} + \frac{3 z}{8} = 0$$

eq10 is the stabilty polynomial of the stencil. The method is absolutely stable if the roots of this polynomial satisfy |xi| < 1, which implies |E(n+1)| < |E(n)|. The boundary of the stability region in the complex z plane is the region for which one of the roots has unit modulus; i.e., xi = exp(I*theta). This defines z = z(theta), which is plotted:

```
> eq11 := subs(xi = exp(I*theta),eq10);
  eq12 := simplify(isolate(eq11,z));
  Z := unapply(rhs(eq12),theta);
  p1 := plot([Re(Z(theta)),Im(Z(theta)),theta=0..2*Pi],axes=boxed):
  p1;
```

$$eq11 := \left(e^{\theta I}\right)^4 + \left(-\frac{55 z}{24} - 1\right)\left(e^{\theta I}\right)^3 + \frac{59}{24}\left(e^{\theta I}\right)^2 z - \frac{37}{24} e^{\theta I} z + \frac{3 z}{8} = 0$$

$$eq12 := z = \frac{24\, e^{3 I \theta}\left(e^{\theta I} - 1\right)}{55\, e^{3 I \theta} - 59\, e^{2 I \theta} + 37\, e^{\theta I} - 9}$$

$$Z := \theta \to \frac{24\, e^{3 I \theta}\left(e^{\theta I} - 1\right)}{55\, e^{3 I \theta} - 59\, e^{2 I \theta} + 37\, e^{\theta I} - 9}$$



Note that this is just the boundary of the stability region; i.e., where the largest root has magnitude one.
We can find the region where the largest root has magnitude less than one by defining:

```
> F := proc(Z):
     if (type(Z,complex)) then:
        max(map(x->abs(x),[fsolve(subs(z=Z,lhs(I*eq10)))]));
     else:
        'procname(args)':
     fi:
  end proc;
```

(16)

$F := \mathbf{proc}(Z)$ **(16)**

    if $type(Z, complex)$ **then**

        $\max(map(x \rightarrow \text{abs}(x), [fsolve(subs(z = Z, lhs(eq10 * \mathrm{I})))]))$

    else

        $'procname(args)'$

    end if

end proc

The argument of the procedure is the complex number z appearing in the stability polynomial eq10. It returns the magnitude of the largest root of the polynomial. We can use contourplot to find the region of the z plane for which F(z) < 1:

```
> p2 := contourplot(F(x+I*y),x=-1..1,y=-1..1,contours=[1],filled=
  true,axes=boxed,grid=[50,50]):
  p2;
```



To reiterate, the first plot indicates where the largest root has magnitude 1 while the second plot show where the largest root has manitude less than 1. Superimposing the two plots gives the entire stability region

```
> display([p1,p2],title=cat(`Stability region of a `,N+1,` stage
  Adams-Bashford stencil`),labels=[`Re(z)`,`Im(z)`]);
```


Stability region of a 4 stage Adams-Bashford stencil
Im(z)
Re(z)
1
0.5
0
−0.5
−1
−1
−0.5
0
0.5
1